

## Só reforçando

- Todas as palavras reservadas são escritas com letras minúsculas.
- O operador de atribuição deverá ser formado pelo sinal < seguido do sinal -, ficando: <- .
- Os identificadores (nome do algoritmo e das variáveis deverão começar por uma **letra**, e os demais caracteres por **letra** ou **algarismo**).
- Os comandos: **imprima**, **leia**, **atribuição** e as **declarações de variáveis** terminam com ; .
- Os comandos **prog** e **fimprog** não têm ; .

## EXERCÍCIOS – LISTA 1

### LEIA, IMPRIMA, ATRIBUIÇÃO E FUNÇÕES

#### algoritmo 28

---

*Imprimir a mensagem: "É PRECISO FAZER TODOS OS ALGORITMOS PARA APRENDER".*

```
prog lea1
  imprima "\nÉ PRECISO FAZER TODOS OS ALGORITMOS PARA APRENDER ";
  imprima "\n";
fimprog
```

---

↪ *imprima "\n";* será sempre colocada para que o prompt não fique na mesma linha da última impressão, uma vez que o cursor não desce automaticamente.

---

#### algoritmo 29

---

*Imprimir seu nome.*

```
prog lea2
  imprima "\n <seu nome>";
  imprima "\n";
fimprog
```

#### algoritmo 30

---

*Criar um algoritmo que imprima o produto entre 28 e 43.*

```
prog lea3
  int produto;
  produto <- 28 * 43;
  imprima "\nO produto entre os dois é: ", produto;
  imprima "\n";
fimprog
```

### algoritmo 31

---

*Criar um algoritmo que imprima a média aritmética entre os números 8, 9 e 7.*

```
prog lea4
  real ma;
  ma <- ( 8 + 9 + 7 ) / 3;
  imprima "\nA media aritmetica e: ", ma;
  imprima "\n";
fimprog
```

### algoritmo 32

---

*Ler um número inteiro e imprimi-lo.*

```
prog lea5
  int num;
  imprima "\n entre com um numero: ";
  leia num;
  imprima "\nnumero : ", num;
  imprima "\n";
fimprog
```

### algoritmo 33

---

*Ler dois números inteiros e imprimi-los.*

```
prog lea6
  int num1, num2;
  imprima "\n entre com um numero: ";
  leia num1;
  imprima "\n entre com outro numero: ";
  leia num2;
  imprima "\nnumero 1 : ", num1;
  imprima "\nnumero 2 : ", num2;
  imprima "\n";
fimprog
```

### algoritmo 34

---

*Ler um número inteiro e imprimir seu sucessor e seu antecessor.*

```
prog lea7
  int numero, suc, ant;
  imprima "\n entre com um numero: ";
  leia numero;
  ant <- numero -1;
  suc <- numero +1;
  imprima "\no sucessor e b", suc,"b o antecessor e b", ant;
  imprima "\n";
fimprog
```

### algoritmo 35

---

*Ler nome, endereço e telefone e imprimi-los.*

```
prog lea8
  string nome, endereco, telefone;
  imprima "\nentre com nome: ";
  leia nome;
  imprima "\nentre com endereco: ";
  leia endereco;
  imprima "\nentre com telefone: ";
  leia telefone;
  imprima "\n\n\n";
  imprima "\nNome      : ", nome;
  imprima "\nEndereco: ", endereco;
  imprima "\nTelefone: ", telefone;
  imprima "\n";
fimprog
```

### algoritmo 36

---

*Ler dois números inteiros e imprimir a soma. Antes do resultado, deverá aparecer a mensagem: Soma.*

```
prog lea9
  int num1, num2, soma;
  imprima "\n entre com um numero: ";
  leia num1;
  imprima "\n entre com outro numero: ";
  leia num2;
  soma <- num1 + num2;
  imprima "\nSoma: ", soma;
  imprima "\n";
fimprog
```

### algoritmo 37

---

*Ler dois números inteiros e imprimir o produto.*

```
prog lea10
  int num1, num2, prod;
  imprima "\n entre com um numero: ";
  leia num1;
  imprima "\n entre com outro numero: ";
  leia num2;
  prod <- num1 * num2;
  imprima "\nproduto: ", prod;
  imprima "\n";
fimprog
```

### algoritmo 38

---

*Ler um número real e imprimir a terça parte deste número.*

```
prog lea11
real num;
imprima "\nentre com um numero com ponto: ";
leia num;
imprima "\na terça parte e: ", num/3;
imprima "\n";
fimprog
```

### algoritmo 39

---

*Entrar com dois números reais e imprimir a média aritmética com a mensagem "média" antes do resultado.*

```
prog lea12
real nota1, nota2, media;
imprima "\ndigite 1a nota: ";
leia nota1;
imprima "\ndigite 2a nota: ";
leia nota2;
media <- ( nota1 + nota2)/2;
imprima "\nmedia: ", media;
imprima "\n";
fimprog
```

### algoritmo 40

---

*Entrar com dois números inteiros e imprimir a seguinte saída:*

dividendo:

divisor:

quociente:

resto:

```
prog lea13
int quoc, rest, val1, val2;
imprima "\nentre com o dividendo: ";
leia val1;
imprima "\nentre com divisor: ";
leia val2;
quoc <- val1 div val2;
rest <- val1 % val2;
imprima "\n\n";
imprima "\ndividendo : ", val1;
imprima "\ndivisor : ", val2;
imprima "\nquociente : ", quoc;
imprima "\nresto : ", rest;
imprima "\n";
fimprog
```

## algoritmo 41

---

*Entrar com quatro números e imprimir a média ponderada, sabendo-se que os pesos são respectivamente: 1, 2, 3 e 4.*

```
prog lea14
  real a, b, c,d, mp;
  imprima "\nentre com 1 numero: ";
  leia a;
  imprima "\nentre com 2 numero: ";
  leia b;
  imprima "\nentre com 3 numero: ";
  leia c;
  imprima "\nentre com 4 numero: ";
  leia d;
  mp <- (a*1 + b*2 + c*3 + d*4)/10;
  imprima "\nmedia ponderada: ", mp;
  imprima "\n";
fimprog
```

## algoritmo 42

---

*Entrar com um ângulo em graus e imprimir: seno, co-seno, tangente, secante, co-secante e co-tangente deste ângulo.*

```
prog lea15
  real angulo, rang;
  imprima "\ndigite um angulo em graus: ";
  leia angulo;
  rang <- angulo*pi /180;
  imprima "\nseno: ", sen(rang);
  imprima "\nco-seno: ", cos(rang);
  imprima "\ntangente: ", tan(rang);
  imprima "\nco-secante: ",1/ sen(rang);
  imprima "\nsecante: ", 1/cos(rang);
  imprima "\ncotangente: ", 1/ tan(rang);
  imprima "\n";
fimprog
```

---

↪ *Alguns ângulos que você digitar poderão não ter resposta em algumas funções, mas este problema será resolvido quando você aprender a estrutura de teste.*

---

## algoritmo 43

---

*Entrar com um número e imprimir o logaritmo desse número na base 10.*

```
prog lea16
  real num, logaritmo;
```

```

imprima "\nentre com o logaritmando: ";
leia num;
logaritmo <- log(num) / log(10);
imprima "\nlogaritmo: ", logaritmo;
imprima "\n";
fimprog

```

#### algoritmo 44

---

*Entrar com o número e a base em que se deseja calcular o logaritmo desse número e imprimi-lo.*

```

prog lea17
real num, base, logaritmo;
imprima "\nentre com o logaritmando: ";
leia num;
imprima "\nentre com a base: ";
leia base;
logaritmo <- log(num) / log(base);
imprima "\no logaritmo de", num, "na base", base, "é:", logaritmo;
imprima "\n";
fimprog

```

#### algoritmo 45

---

*Entrar com um número e imprimir a seguinte saída:*

```

numero:
quadrado:
raiz quadrada:
prog lea18
real num, quad, raizquad;
imprima "\ndigite numero: ";
leia num;
quad <- num ** 2;
raizquad <- raiz(num);
imprima "\nnumero: ", num;
imprima "\nquadrado: ", quad;
imprima "\nraiz quadrada: ", raizquad;
imprima "\n";
fimprog

```

#### algoritmo 46

---

*Fazer um algoritmo que possa entrar com o saldo de uma aplicação e imprima o novo saldo, considerando o reajuste de 1%.*

```

prog lea19
real saldo, nsaldo;
imprima "\ndigite saldo: ";

```

```

leia saldo;
nsaldo <-saldo * 1.01;
imprima "\nnovo saldo: ",nsaldo;
imprima "\n";
fimprog

```

### algoritmo 47

*Entrar com um número no formato CDU e imprimir invertido: UDC. (Exemplo: 123, sairá 321.) O número deverá ser armazenado em outra variável antes de ser impresso.*

```

prog lea20
int num, c, d, u, num1;
imprima "\nentre com um número de 3 dígitos: ";
leia num;
c <- num div 100;
d <- num % 100 div 10;
u <- num % 10;
num1 <- u*100 + d*10 + c;
imprima "\nnúmero: ", num;
imprima "\ninvertido: ", num1;
imprima "\n";
fimprog

```

### algoritmo 48

*Antes de o racionamento de energia ser decretado, quase ninguém falava em quilowatts; mas, agora, todos incorporaram essa palavra em seu vocabulário. Sabendo-se que 100 quilowatts de energia custa um sétimo do salário mínimo, fazer um algoritmo que receba o valor do salário mínimo e a quantidade de quilowatts gasta por uma residência e calcule. Imprima:*

- o valor em reais de cada quilowatt
- o valor em reais a ser pago
- o novo valor a ser pago por essa residência com um desconto de 10%.

```

prog lea21
real sm, qtdade, preco, vp, vd;
imprima "\nentre com o salário mínimo: ";
leia sm;
imprima "\nentre com a quantidade em quilowatt: ";
leia qtdade;
# divide por 7 para achar o preço de 100 Kw e por 100 para achar de 1 Kw
preco <- sm /700;
vp <- preco * qtdade;
vd <- vp * 0.9;
imprima "\npreço do quilowatt: ", preco, "\n valor a ser pago: ", vp,

```

```

"\n valor com desconto: ", vd;
imprima "\n";
fimprog

```

## algoritmo 49

---

*Entrar com um nome e imprimir:*

```

todo nome:
primeiro caractere:
ultimo caractere:
do primeiro ate o terceiro:
quarto caractere:
todos menos o primeiro:
os dois ultimos:
prog lea22
string nome;
int n;
imprima "\nentre com nome: ";
leia nome;
imprima "\ntodo nome: " , nome;
imprima "\nprimeiro caractere: ", strprim(nome);
imprima "\nultimo caractere: ", strult(nome);
imprima "\nprimeiro ao terceiro caractere: ", strnprim(nome,3);
imprima "\nquarto caractere: ", strelem(nome,3);
imprima "\ntodos menos o primeiro: ", strresto ( nome);
n <-strtam(nome) -2;
imprima "\nos dois ultimos: " , strnresto(nome,n);
imprima "\n";
fimprog

```

## algoritmo 50

---

*Entrar com a base e a altura de um retângulo e imprimir a seguinte saída:*

```

perimetro:
area:
diagonal:
prog lea23
real perimetro, area, diagonal, base, altura;
imprima "\ndigite base: ";
leia base;
imprima "\ndigite altura: ";
leia altura;
perimetro <- 2*(base + altura);
area <-base * altura;
diagonal <- raiz(base**2 + altura**2);
imprima "\nperimetro = " ,perimetro;
imprima "\narea = " , area ;

```



```

    imprima "\ndiagonal = ", diagonal ;
    imprima "\n";
fimprog

```

## algoritmo 51

---

*Entrar com o raio de um círculo e imprimir a seguinte saída:*

```

perimetro:
area:
prog lea24
real raio, perimetro, area;
imprima "\ndigite raio: ";
leia raio;
perimetro <- 2* pi * raio;
area <- pi * raio ** 2;
imprima "\nperimetro : " , perimetro;
imprima "\narea : " , area ;
imprima "\n";
fimprog

```

## algoritmo 52

---

*Entrar com o lado de um quadrado e imprimir:*

```

perimetro:
area:
diagonal:
prog lea25
real lado, perimetro, area, diagonal;
imprima "\ndigite o lado do quadrado: ";
leia lado ;
perimetro <- 4 * lado;
area <- lado ** 2;
diagonal <- lado * raiz(2);
imprima "\nperimetro: " , perimetro;
imprima "\narea: " , area;
imprima "\ndiagonal: " , diagonal;
imprima "\n";
fimprog

```

## algoritmo 53

---

*Entrar com os lados a, b, c de um paralelepípedo. Calcular e imprimir a diagonal.*

```

prog lea26
real a, b, c, diagonal;
imprima "\nentre com a base: ";
leia a;
imprima "\nentre com a altura: ";
leia b;

```

```

imprima "\nentre com a profundidade: ";
leia c;
diagonal <-raiz( a**2 + b**2 + c**2 );
imprima "\ndiagonal : ", diagonal;
imprima "\n";
fimprog

```

## algoritmo 54

---

*Criar um algoritmo que calcule e imprima a área de um triângulo.*

```

prog lea27
real a, b;
imprima "\nEntre com a base: ";
leia a;
imprima "\nEntre a altura do um triângulo: ";
leia b;
imprima "\nArea = ", (a * b)/2;
imprima "\n";
fimprog

```

## algoritmo 55

---

*Criar um algoritmo que calcule e imprima a área de um losango.*

```

prog lea28
real diagmaior, diagmenor, area;
imprima "\nmedida da diagonal maior: ";
leia diagmaior;
imprima "\nmedida da diagonal menor: ";
leia diagmenor;
area <- (diagmaior * diagmenor)/2;
imprima "\narea =", area;
imprima "\n";
fimprog

```

## algoritmo 56

---

*Entrar com nome e idade. Imprimir a seguinte saída:*

```

nome:
idade:
prog lea29
string nome;
int idade;
imprima "\ndigite nome: ";
leia nome;
imprima "\ndigite idade: ";
leia idade;
# a linha abaixo é para dar uma separação entre a entrada e a saída

```

```

imprima "\n\n";
imprima "\nnome = ", nome;
imprima "\nidade = ", idade;
imprima "\n";
fimprog

```

## algoritmo 57

*Entrar com as notas da PR1 e PR2 e imprimir a média final:*

- *truncada:*
- *arredondada:*

```

prog lea30
real pr1, pr2, mf;
imprima "\ndigite pr1: ";
leia pr1;
imprima "\ndigite pr2: ";
leia pr2;
mf <- ( pr1 + pr2 ) / 2;
imprima "\nmedia truncada = ", realint((mf- 0.5)+0.001);
imprima "\nmedia arredondada = ", realint( mf+0.001);
imprima "\n";
fimprog

```

VÍDEO	
digite pr1:2.3	digite pr1:7.9
digite pr2:3.7	digite pr2:8.1
media truncada = 3 media arredondada = 3	media truncada = 8 media arredondada = 8
digite pr1:2.8	digite pr1:6.9
digite pr2:2.7	digite pr2:8.1
media truncada = 2 media arredondada = 3	media truncada = 7 media arredondada = 8

## algoritmo 58

*Entrar com valores para xnum1, xnum2 e xnum3 e imprimir o valor de x, sabendo-se que:*

$$X = xnum1 + \frac{xnum2}{xnum3 + xnum1} + 2(xnum1 - xnum2) + \log_2^{64}$$

```

prog lea31
real xnum1, xnum2, xnum3, x;
imprima "\nEntrar com 1 valor: ";
leia xnum1;
imprima "\nEntrar com 2 valor: ";
leia xnum2;
imprima "\nEntrar com 3 valor: ";
leia xnum3;
x <- xnum1 + xnum2 / (xnum3 + xnum1) + 2 *(xnum1 - xnum2) + log(64.)/
log(2.);
imprima "\nX = ", x;
imprima "\n";
fimprog

```

### algoritmo 59

---

*Entrar com os valores dos catetos de um triângulo retângulo e imprimir a hipotenusa.*

```

prog lea32
real a,b,c;
imprima "\nEntrar com 1 cateto: ";
leia b;
imprima "\nEntrar com 2 cateto: ";
leia c;
a <- raiz (b**2 + c**2);
imprima "\nA hipotenusa e: ", a;
imprima "\n";
fimprog

```

### algoritmo 60

---

*Entrar com a razão de uma PA e o valor do 1º termo. Calcular e imprimir o 10º termo da série.*

```

prog lea33
int dec, razao, termo;
imprima "\nEntrar com o 1o termo: ";
leia termo;
imprima "\nEntrar com a razao: ";
leia razao;
dec <- termo + 9* razao;
imprima "\nO 10 termo desta P.A. e: ", dec;
imprima "\n";
fimprog

```

## algoritmo 61

---

*Entrar com a razão de uma PG e o valor do 1º termo. Calcular e imprimir o 5º termo da série.*

```
prog lea34
  int quinto, razao, termo;
  imprima "\nEntre com o 1o termo: ";
  leia termo;
  imprima "\nEntre com a razao: ";
  leia razao;
  quinto <- termo * razao ^4;
  imprima "\nO 5o termo desta P.G. e: ", quinto;
  imprima "\n";
fimprog
```

## algoritmo 62

---

*Em épocas de pouco dinheiro, os comerciantes estão procurando aumentar suas vendas oferecendo desconto. Faça um algoritmo que possa entrar com o valor de um produto e imprima o novo valor tendo em vista que o desconto foi de 9%.*

```
prog lea35
  real preco, npreco;
  imprima "\ndigite valor do produto: ";
  leia preco;
  npreco <- preco * 0.91;
  imprima "\npreco com desconto: ", npreco;
  imprima "\n";
fimprog
```

## algoritmo 63

---

*Criar um algoritmo que efetue o cálculo do salário líquido de um professor. Os dados fornecidos serão: valor da hora aula, número de aulas dadas no mês e percentual de desconto do INSS.*

```
prog lea36
  int na;
  real vha, pd, td, sb, sl;
  imprima "\nhoras trabalhadas: ";
  leia na ;
  imprima "\nvalor da hora-aula: ";
  leia vha ;
  imprima "\npercentual de desconto: ";
  leia pd ;
  sb <- na * vha;
  td <- ( pd / 100) * sb;
  sl <- sb - td;
```

```

imprima "\nsalario liquido: ",sl;
imprima "\n";
fimprog

```

#### algoritmo 64

---

*Ler uma temperatura em graus centígrados e apresentá-la convertida em graus Fahrenheit. A fórmula de conversão é:  $F = \frac{9 \cdot C + 160}{5}$  onde F é a temperatura em Fahrenheit e C é a temperatura em centígrados.*

```

prog lea37
real f, c;
imprima "\ndigite o valor da temperatura em graus centígrados: ";
leia c;
f <- ( 9 * c + 160)/5;
imprima "\no valor da temperatura em graus fahrenheit e =", f;
imprima "\n";
fimprog

```

#### algoritmo 65

---

*Calcular e apresentar o valor do volume de uma lata de óleo, utilizando a fórmula:  $\text{volume} = 3.14159 * R^2 * \text{altura}$ .*

```

prog lea38
real volume, altura, raio;
imprima "\ndigite a altura da lata: ";
leia altura;
imprima "\ndigite o raio da lata: ";
leia raio;
volume <- pi *raio ** 2 *altura;
imprima "\no volume da lata e = ", volume;
imprima "\n";
fimprog

```

#### algoritmo 66

---

*Efetuar o cálculo da quantidade de litros de combustível gastos em uma viagem, sabendo-se que o carro faz 12 km com um litro. Deverão ser fornecidos o tempo gasto na viagem e a velocidade média.*

Utilizar as seguintes fórmulas:

$$\text{distância} = \text{tempo} \times \text{velocidade.}$$

$$\text{litros usados} = \text{distância} / 12.$$

*O algoritmo deverá apresentar os valores da velocidade média, tempo gasto na viagem, distância percorrida e a quantidade de litros utilizados na viagem.*

```

prog lea39
real tempo, vel, dist, litros;
imprima "\ndigite o tempo gasto: ";
leia tempo;
imprima "\ndigite a velocidade media: ";
leia vel;
dist <- tempo * vel;
litros <- dist / 12;
imprima "\nvelocidade = ", vel, "\ntempo = ", tempo, "\ndistancia = ",
dist, "\nlitros = ", litros;
imprima "\n";
fimprog

```

### algoritmo 67

---

*Efetuar o cálculo do valor de uma prestação em atraso, utilizando a fórmula:  
prestação = valor + (valor\*(taxa/100)\*tempo).*

```

prog lea40
real prest, valor, taxa;
int tempo;
imprima "\ndigite o valor da prestação: ";
leia valor;
imprima "\ndigite a taxa: ";
leia taxa;
imprima "\ndigite o tempo(numero de meses): ";
leia tempo;
prest <- valor+(valor*(taxa/100)*tempo);
imprima "\no valor da prestacao em atraso e =", prest;
imprima "\n";
fimprog

```

### algoritmo 68

---

*Ler dois valores para as variáveis A e B, efetuar a troca dos valores de forma que a variável A passe a ter o valor da variável B e que a variável B passe a ter o valor da variável A. Apresentar os valores trocados.*

```

prog lea41
real a, b, aux;
imprima "\ndigite 1 numero com ponto: ";
leia a;
imprima "\ndigite 2 numero com ponto: ";
leia b;
aux <- a;
a <- b;
b <-aux;
imprima "\na = ", a, "\nb = ", b;

```

```
imprima "\n";  
fimprog
```

### algoritmo 69

---

*Criar um algoritmo que leia o numerador e o denominador de uma fração e transformá-lo em um número decimal.*

```
prog lea42  
int num, denom;  
imprima "\ndigite numerador: ";  
leia num;  
imprima "\ndigite denominador: ";  
leia denom;  
imprima "\ndecimal: ", num / denom;  
imprima "\n";  
fimprog
```

### algoritmo 70

---

*Todo restaurante, embora por lei não possa obrigar o cliente a pagar, cobra 10% para o garçom. Fazer um algoritmo que leia o valor gasto com despesas realizadas em um restaurante e imprima o valor total com a gorjeta.*

```
prog lea43  
real cres, cgorj;  
imprima "\nEntre com o valor da conta: ";  
leia cres;  
cgorj <- cres * 1.1;  
imprima "\nO valor da conta com a gorjeta sera: ", formatar(cgorj,2);  
imprima "\n";  
fimprog
```

### algoritmo 71

---

*Criar um algoritmo que leia um valor de hora e informe quantos minutos se passaram desde o início do dia.*

```
prog lea44  
int hora, tminuto, minuto;  
imprima "\nentre com hora atual: ";  
leia hora;  
imprima "\nentre com minutos: ";  
leia minuto;  
tminuto <- hora * 60 + minuto;  
imprima "\nAte agora se passaram: ", tminuto, " minutos";  
imprima "\n";  
fimprog
```



## algoritmo 72

---

*Criar um algoritmo que leia o valor de um depósito e o valor da taxa de juros. Calcular e imprimir o valor do rendimento e o valor total depois do rendimento.*

```
prog lea45
  real deposito, taxa, valor, total;
  imprima "\nentre com depósito: ";
  leia deposito;
  imprima "\nentre coma taxa de juros: ";
  leia taxa;
  valor <- deposito*taxa/100;
  total <- deposito + valor;
  imprima "\nRendimentos: ", valor, "\nTotal: ", total;
  imprima "\n";
fimprog
```

## algoritmo 73

---

*Criar um algoritmo que receba um número real, calcular e imprimir:*

- a parte inteira do número
- a parte fracionária do número
- o número arredondado

```
prog lea46
  real num, numfrac;
  int numi, numa;
  imprima "\nentre com um numero com parte fracionaria: ";
  leia num;
  numi <- realint((num - 0.5));
  numfrac <- num - numi;
  numa <- realint(num + 0.00001);
  imprima "\nparte inteira: ", numi;
  imprima "\nparte fracionaria: ", formatar((numfrac + 0.00001), 3);
  imprima "\nnumero arredondado: ", numa;
  imprima "\n";
fimprog
```

---

↪ *Qualquer dúvida, consulte o Apêndice I.*

---

VÍDEO	
entre com um numero com parte fracionaria:7.1	entre com um numero com parte fracionaria:8.4999
parte inteira:7	parte inteira:8
parte fracionária:0.100	parte fracionária:0.499
numero arredondado:7	numero arredondado:8
entre com um numero com parte fracionaria:8.5	entre com um numero com parte fracionaria:7.4999
parte inteira:8	parte inteira:7
parte fracionária:0.500	parte fracionária:0.499
numero arredondado:9	numero arredondado:7
entre com um numero com parte fracionaria:7.49	entre com um numero com parte fracionária:8.0
parte inteira:7	parte inteira:8
parte fracionária:0.490	parte fracionária:1.000
numero arredondado:7	numero arredondado:8

#### algoritmo 74

*Para vários tributos, a base de cálculo é o salário mínimo. Fazer um algoritmo que leia o valor do salário mínimo e o valor do salário de uma pessoa. Calcular e imprimir quantos salários mínimos ela ganha.*

```

prog lea47
real salmin, salpe, num;
imprima "\nentre com o salario minimo: ";
leia salmin;
imprima "\nentre com o salario da pessoa: ";
leia salpe;
num <- salpe / salmin;
imprima "\na pessoa ganha ", num, " salarios minimos";
imprima "\n";
fimprog

```

#### algoritmo 75

*Criar um algoritmo que leia o peso de uma pessoa, só a parte inteira, calcular e imprimir:*

- o peso da pessoa em gramas
- novo peso, em gramas, se a pessoa engordar 12%

```

prog lea48
  int peso, pesogramas, novopeso;
  imprima "\nentre com seu peso, só a parte inteira: ";
  leia peso;
  pesogramas <- peso * 1000;
  novopeso <- pesogramas * 1.12;
  imprima "\n peso em gramas: ", pesogramas;
  imprima "\nnovo peso: ", novopeso;
  imprima "\n";
fimprog

```

### algoritmo 76

---

*Criar um algoritmo que leia um número entre 0 e 60 e imprimir o seu sucessor, sabendo que o sucessor de 60 é 0. Não pode ser utilizado nenhum comando de seleção e nem de repetição.*

```

prog leia49
  int num;
  imprima "\ndigite numero: ";
  leia num;
  imprima "\nsucessor: ", (num + 1) % 61;
  imprima "\n";
fimprog

```

### algoritmo 77

---

*Ler dois números reais e imprimir o quadrado da diferença do primeiro valor pelo segundo e a diferença dos quadrados.*

```

prog lea50
  real a, b, d, q;
  imprima "\ndigite 1 numero: ";
  leia a;
  imprima "\ndigite 2 numero: ";
  leia b;
  d <- (a - b)**2;
  q <- a**2 - b**2;
  imprima "\no quadrado da diferenca =", d , "\ndiferenca dos quadrados =", q;
  imprima "\n";
fimprog

```

### algoritmo 78

---

*Dado um polígono convexo de  $n$  lados, podemos calcular o número de diagonais diferentes ( $nd$ ) desse polígono pela fórmula :  $nd = n(n - 3) / 2$ . Fazer um algoritmo que leia quantos lados tem o polígono, calcule e escreva o número de diagonais diferentes ( $nd$ ) do mesmo.*

```

prog lea51
real nd;
int n;
imprima "\ndigite o numero de lados do poligono: ";
leia n;
nd <- n * ( n - 3) / 2;
imprima "\nnumero de diagonais: ", nd;
imprima "\n";
fimprog

```

## algoritmo 79

*Uma pessoa resolveu fazer uma aplicação em uma poupança programada. Para calcular seu rendimento, ela deverá fornecer o valor constante da aplicação mensal, a taxa e o número de meses. Sabendo-se que a fórmula usada para este cálculo é:*

$$\text{valor acumulado} = P * \frac{(1+i)^n - 1}{i}$$

*i = taxa  
P = aplicação mensal  
n = número de meses*

```

prog lea52
real va, i, p;
int n;
imprima "\ndigite o valor da aplicacao: ";
leia p;
imprima "\ndigite a taxa( 0 - 1): ";
leia i;
imprima "\ndigite o numero de meses: ";
leia n;
va <- p * (((1 + i) ** n) - 1) / i;
imprima "\nO valor acumulado: ", va;
imprima "\n";
fimprog

```

## algoritmo 80

*Criar um algoritmo que leia a quantidade de fitas que uma locadora de vídeo possui e o valor que ela cobra por cada aluguel, mostrando as informações pedidas a seguir:*

- *Sabendo que um terço das fitas são alugadas por mês, exiba o faturamento anual da locadora;*
- *Quando o cliente atrasa a entrega, é cobrada uma multa de 10% sobre o valor do aluguel. Sabendo que um décimo das fitas alugadas no mês são devolvidas com atraso, calcule o valor ganho com multas por mês;*
- *Sabendo ainda que 2% de fitas se estragam ao longo do ano, e um décimo do total é comprado para reposição, exiba a quantidade de fitas que a locadora terá no final do ano.*

```

prog lea53
  int quant;
  real valAluguel, fatAnual, multas, quantFinal;
  imprima "\n Digite a quantidade de fitas: ";
  leia quant;
  imprima "\n Digite o valor do aluguel: ";
  leia valAluguel;
  fatAnual <- quant/3 * valAluguel * 12;
  imprima "\n Faturamento anual: ", fatAnual;
  multas <- valAluguel * 0.1 * (quant/3)/10;
  imprima "\n Multas mensais: ", multas;
  quantFinal <- quant - quant * 0.02 + quant/10; /* quant * 1.08 */
  imprima "\n Quantidade de fitas no final do ano : ", quantFinal;
  imprima "\n";
fimprog

```

## algoritmo 81

---

*Criar um algoritmo que, dado um número de conta corrente com três dígitos, retorne o seu dígito verificador, o qual é calculado da seguinte maneira:*

*Exemplo: número da conta: 235*

- Somar o número da conta com o seu inverso:  $235 + 532 = 767$
- multiplicar cada dígito pela sua ordem posicional e somar estes resultados: 7 6 7

$$\begin{array}{r}
 7 \quad 6 \quad 7 \\
 \times 1 \quad \times 2 \quad \times 3 \\
 \hline
 7 + 12 + 21 = 40
 \end{array}$$

- o último dígito desse resultado é o dígito verificador da conta ( $40 \rightarrow 0$ ).

```

prog lea54
  int conta, inv, digito, d1, d2, d3, soma;
  imprima "\n Digite conta de tres digitos: ";
  leia conta;
  d1 <- conta div 100;
  d2 <- conta % 100 div 10;
  d3 <- conta % 100 % 10;
  inv <- d3 * 100 + d2 * 10 + d1;
  soma <- conta + inv;
  d1 <- (soma div 100) * 1;
  d2 <- (soma % 100 div 10) * 2;
  d3 <- (soma % 100 % 10) * 3;
  digito <- (d1 + d2 + d3) % 10;
  imprima "\n digito verificador: ", digito;
  imprima "\n";
fimprog

```